ME 406
Examples of Periodic Solutions

In[403]:=
```
sysid
```

```
Mathematica  6.0.3, DynPac 11.02, 2/6/2009
```

In[404]:=
```
plotreset; intreset; imsize = 250;
```

■ Introduction

In this notebook we look at several examples of periodic solutions of autonomous systems of two equations. We visualize the solutions in the phase plane, where periodic solutions have a distinctive topological signature: their orbits are simple closed curves. As we shall see from the examples, there are a variety of possible outcomes from the search for periodic solutions. For some systems there are no periodic solutions. For others, every solution is periodic. For still others, there are periodic solutions in some regions of the phase plane but not everywhere. Finally there is the surprising and interesting phenomenon of the limit cycle, which is an isolated periodic solution. Our examples will illustrate all of this behavior. In later notebooks we will cover some of the tools used here in a more systematic search for periodic solutions.

■ Example 1: Linear System with All Solutions Periodic

We have already used the linear oscillator a number of times as an example, and we know that the undamped oscillator has periodic solutions. We repeat that example here. In scaled form the equations are

$$\dot{x} = y, \;\; \dot{y} = -x \;\; .$$

It is easy to solve the initial value problem for this system analytically. If we let $x(0) = a$, and $\dot{x}(0) = b$, we get the solution

$$x(t) = a\cos(t) + b\sin(t), \;\; y(t) = -a\sin(t) + b\cos(t) \; .$$

Thus all solutions are periodic with the same period, namely 2π. From the above solution, we find by direct calculation that $[x(t)]^2 + [y(t)]^2 = a^2 + b^2$, so that all of the orbits are circles.

Of course if we didn't know any of this, we could find a numerical approximation to the truth by solving the system numerically and plotting the orbits. Let's do that. We define the system for DynPac.

In[405]:=
```
setstate[{x, y}]; setparm[{}]; slopevec = {y, -x};
```

In[406]:=
```
sysname = "LinOsc";
```

We now construct a single solution, with initial point {1,0}.

In[407]:=
```
t0 = 0.0; h = 0.02; nsteps = 350; initvec = {1, 0};
```

In[408]:=
```
sol1 = integrate[initvec, t0, h, nsteps];
```

In[409]:=
```
plrange = {{-2, 2}, {-2, 2}}; asprat = 1.0;
```

In[410]:=
```
arrowflag = True; arrowvec = {1 / 3, 3 / 4};
```

In[411]:=
```
graph1 = phaser[sol1]
```

Out[411]=



We see that the orbit is a simple closed curve. How do we find the period of this numerical solution? We use the function period, applied to the solution.

In[412]:=
```
period[sol1]
```

Out[412]=
```
6.28
```

This function returns an answer accurate to within one time step, and we see that the result is consistent with the exact answer of 2π.

We construct several other solutions, find their periods and then plot them together.

```
In[413]:=
      sol2 = integrate[{0.5, 0}, t0, h, nsteps];

In[414]:=
      period[sol2]

Out[414]=
      6.28

In[415]:=
      sol3 = integrate[{1.5, 0}, t0, h, nsteps];

In[416]:=
      period[sol3]

Out[416]=
      6.28

In[417]:=
      sol4 = integrate[{2.0, 0}, t0, h, nsteps];

In[418]:=
      period[sol4]

Out[418]=
      6.28

In[419]:=
      phaser[{sol1, sol2, sol3, sol4}]

Out[419]=
```



The numerical work we have just finished suggests that all solutions are periodic and that they have the same period, 6.28. Although we can never **prove** such results with numerical work, the numerical techniques are our primary exploratory tools, and at the very least suggest what properties we might want to establish by analysis.

It is worth noting that this system, which has all solutions periodic, is immediately converted to a system with no periodic solutions if we have any amount of damping, no matter how small. There is an important concept called structural stability behind that statement. We will discuss it later in the course.

■ Example 2: Nonlinear System with All Solutions Periodic

For our next example, we consider a nonlinear oscillator governed by the well-known Duffing's equation. We consider the simplest case of no damping. The equation is

$$\ddot{x} + x + x^3 = 0 \ .$$

A physical model for this system is an oscillator with a nonlinear spring, for which the restoring spring force is given by $x + x^3$. In vibration theory such a spring is called a hard spring, because the spring force per unit extension, $1 + 3x^2$, increases as the displacement increases. We convert the equation to a system.

$$\dot{x} = y \ , \ \dot{y} = -x - x^3.$$

It is actually possible to solve this equation analytically in terms of elliptic functions. Although we won't carry this out, we will look at the first step. The second order equation is of a form known as a conservative equation. It admits a conservation law which is an energy equation. The general form of a conservation equation is

$$\ddot{x} + f(x) = 0 \ .$$

If we multiply through by $\dot{x}$, we can integrate the result to get

$$\frac{1}{2}\dot{x}^2 + V(x) = \text{constant} \ ,$$

where the potential V is given by $\int f(x)\,dx$. For the Duffing equation, the result is

$$\frac{1}{2}\dot{x}^2 + \frac{1}{2}x^2 + \frac{1}{4}x^4 = \frac{1}{2}x_0^{\,2} + \frac{1}{4}x_0^{\,4} \ ,$$

where we have parametrized the solution by initial conditions $x(0) = x_0$, $\dot{x}(0) = 0$. We may solve for $\dot{x}$ in terms of x:

$$\dot{x} = \sqrt{\left(x_0^{\,2} - x^2\right)\left(1 + \frac{1}{2}\left[x_0^{\,2} + x^2\right]\right)} \ .$$

The above equation is separable and can be further manipulated to generate the solution in terms of elliptic functions. We will use it in a more limited way shortly to derive an expression for the period of the motion.

We define this system for DynPac, generate four solutions, plot them to discover that they are periodic, and then calculate the periods.

In[420]:=

```
setstate[{x, y}]; setparm[{}]; slopevec = {y, -x - x^3};
```

```
In[421]:=
        sysname = "Duffing";
```

We generate a solution with x_0= 1 and then plot it.

```
In[422]:=
        sol1 = integrate[{1, 0}, 0.0, 0.02, 400];
```

```
In[423]:=
        period[sol1]
Out[423]=
        4.76
```

```
In[424]:=
        arrowvec = {1 / 3};
```

```
In[425]:=
        plrange = {{-3, 3}, {-3, 3}};
```

```
In[426]:=
        phaser[sol1]
Out[426]=
```



We see a closed curve for the periodic orbit, but it is not circular. In the limit of small amplitude the cubic nonlinearity in the force should be unimportant, the orbit should be close to circular, and the period should be close to 2π. Let's check that by doing a second integration for x_0= 0.25.

```
In[427]:=
        sol2 = integrate[{0.25, 0}, 0.0, 0.02, 400];
```

In[428]:=
```
period[sol2]
```
Out[428]=
```
6.14
```
In[429]:=
```
phaser[sol2]
```
Out[429]=



As we predicted the orbit is close to circular and the period is close to 2π. Now let's look at a larger amplitude.

In[430]:=
```
sol3 = integrate[{1.5, 0}, 0.0, 0.02, 400];
```
In[431]:=
```
period[sol3]
```
Out[431]=
```
3.86
```

In[432]:=
```
phaser[sol3]
```

Out[432]=



One more solution.

In[433]:=
```
sol4 = integrate[{1.75, 0}, 0.0, 0.02, 400];
```

In[434]:=
```
period[sol4]
```

Out[434]=
```
3.5
```

In[435]:=

```
phaser[sol4]
```

Out[435]=



We put all of these together.

In[436]:=
```
phaser[{sol1, sol2, sol3, sol4}]
```

Out[436]=



Notice also the strong dependance of the period on the amplitude. The larger amplitude motions have shorter periods. This is a consequence of the nonlinear spring force. In the larger amplitude motions, the restoring force per unit extension is much larger because of the cubic term in the force.

Let's return to the equation derived above for $\dot{x}$ in terms of x to calculate the period of the motion. We separate the variables and integrate from $x = 0$ to $x = x_0$. This is one-quarter of the way around the orbit, and by symmetry gives us 1/4 of the period. Hence the period tau as a function of amplitude x0 is given by

In[437]:=
```
tau[x0_] := 4 * N[Integrate[1 / Sqrt[(x0^2 - x^2) (1 + 1/2 (x0^2 + x^2))], {x, 0, x0}]]
```

We try this out for the four orbits we calculated.

In[438]:=
```
tau[1 / 4]
```

Out[438]=
```
6.14116
```

In[439]:=
```
tau[1]
```

Out[439]=
```
4.76802
```

```
In[440]:=
      tau[3 / 2]

Out[440]=
      3.86497

In[441]:=
      tau[7 / 4]

Out[441]=
      3.49648
```

These values are quite close to the values computed earlier with the function period.

▪ Example 3: Nonlinear System with Some Solutions Periodic

We continue our study of Duffing's equation, but now with a soft spring:

$$\ddot{x} + x - x^3 = 0 \ .$$

With the minus sign on the cubic term, the spring force per unit extension decreases with increasing amplitude. In fact the spring force goes to zero at $x = \pm 1$, and becomes negative for $|x| > 1$. This suggests that not all solutions will be periodic, because there is no restoring force for $|x| > 1$.

We define the system for DynPac.

```
In[442]:=
      setstate[{x, y}]; setparm[{}]; slopevec = {y, -x + x^3};

In[443]:=
      sysname = "Soft Duffing";
```

In the previous hard spring example, we didn't discuss equilibrium points, because it wasn't really necessary for understanding the results. The only equilibrium point in that case was the rest state at the origin. In the present case, there are three equilbria:

```
In[444]:=
      findpolyeq

Out[444]=
      {{-1, 0}, {0, 0}, {1, 0}}
```

The equilibria at $x = \pm 1$ play an important role in the phase portrait. Let's examine those equilibria first.

```
In[445]:=
      classify2D[{1, 0}]

      unstable - saddle

In[446]:=
      classify2D[{-1, 0}]

      unstable - saddle
```

Thus they are both saddle points. Now the equilibrium at the origin:

In[447]:=
```
classify2D[{0, 0}]
```

```
stable (L), indeterminate (NL) - center
```

A center as we might have expected.

We start by looking at an orbit fairly near the origin, so that the cubic term will be relatively unimportant. We expect that orbit to look like a linear oscillator orbit with a period on the order of 2π.

In[448]:=
```
sol1 = integrate[{0.25, 0}, 0.0, 0.02, 400];
```

In[449]:=
```
period[sol1]
```

Out[449]=
```
6.44
```

In[450]:=
```
setcolor[{Blue}];
```

In[451]:=
```
graph1 = phaser[sol1]
```

Out[451]=



Now we try a slightly larger orbit.

In[452]:=
```
sol2 = integrate[{0.5, 0}, 0.0, 0.02, 400];
```

In[453]:=
```
period[sol2]
```
Out[453]=
```
6.98
```
In[454]:=
```
graph2 = phaser[sol2]
```
Out[454]=



The orbit is a little flattened and the period is greater. We look at a still larger orbit.

In[455]:=
```
sol3 = integrate[{0.75, 0}, 0.0, 0.02, 600];
```
In[456]:=
```
period[sol3]
```
Out[456]=
```
8.36
```

In[457]:=
```
graph3 = phaser[sol3]
```

Out[457]=



Now the flattening is very noticeable, and the period is considerably longer.

Now we look at a considerably larger orbit, and we turn on range checking to prevent a blowup.

In[458]:=
```
rangeflag = True; ranger = plrange;
```

In[459]:=
```
bothdirflag = True;
```

In[460]:=
```
sol4 = integrate[{1.5, 0}, 0.0, 0.02, 500];
```

In[461]:=
```
sol5 = integrate[{-1.5, 0}, 0.0, 0.02, 500];
```

In[462]:=
```
setcolor[{Green}];
```

In[463]:=
```
graph4 = phaser[{sol4, sol5}]
```

Out[463]=



Now we have lost our periodic solutions, which we might have expected, because we are in the range where the spring force is no longer restoring. We look at four more orbits in this range.

In[464]:=
```
sol6 = integrate[{2.5, 0}, 0.0, 0.02, 500];
```

In[465]:=
```
sol7 = integrate[{-2.5, 0}, 0.0, 0.02, 500];
```

In[466]:=

```
graph5 = phaser[{sol6, sol7}]
```

Out[466]=



In[467]:=

```
sol8 = integrate[{0, 1.5}, 0.0, 0.02, 500];
```

In[468]:=

```
sol9 = integrate[{0, -1.5}, 0.0, 0.02, 500];
```

In[469]:=

```
graph6 = phaser[{sol8, sol9}]
```

Out[469]=



Given that there are periodic solutions in some parts of the phase plane but not others, it becomes important to find the boundary between those two regions. We might suspect that the saddle points at $x = \pm 1$ have something to do with this, especially since these values of x are where the spring force changes character. Let's construct the solutions through the saddle points to see what we get. We start by finding the eigenvectors at each point.

In[470]:=

```
eigleft = eigsys[{-1, 0}]
```

Out[470]=

$$\left\{\left\{-\sqrt{2}, \sqrt{2}\right\}, \left\{\left\{-\frac{1}{\sqrt{2}}, 1\right\}, \left\{\frac{1}{\sqrt{2}}, 1\right\}\right\}\right\}$$

In[471]:=

```
eigright = eigsys[{1, 0}]
```

Out[471]=

$$\left\{\left\{-\sqrt{2}, \sqrt{2}\right\}, \left\{\left\{-\frac{1}{\sqrt{2}}, 1\right\}, \left\{\frac{1}{\sqrt{2}}, 1\right\}\right\}\right\}$$

In[472]:=

```
eigleft1 = eigleft[[2, 1]]
```

Out[472]=

$$\left\{-\frac{1}{\sqrt{2}}, 1\right\}$$

In[473]:=
```
eigleft2 = eigleft[[2, 2]]
```

Out[473]=

$$\left\{\frac{1}{\sqrt{2}}, 1\right\}$$

In[474]:=
```
eigright1 = eigright[[2, 1]]
```

Out[474]=

$$\left\{-\frac{1}{\sqrt{2}}, 1\right\}$$

In[475]:=
```
eigright2 = eigright[[2, 2]]
```

Out[475]=

$$\left\{\frac{1}{\sqrt{2}}, 1\right\}$$

In[476]:=
```
eqleft = {-1, 0}; eqright = {1, 0};
```

Now we use these eigenvectors to construct a useful set of initial conditions. Each condition in the set is a point displaced a small amplitude eps from the equilibrium along the direction of the eigenvector. Because some of the integral curves actually connect the two saddle points (a statement based on hindsight from the already constructed picture!) we adjust our choice of initial conditions so that these curves are not computed twice.

In[477]:=
```
eps = 0.01;
```

In[478]:=
```
initset = {eqleft + eps * eigleft1, eqleft - eps * eigleft1, eqleft + eps * eigleft2,
   eqleft - eps * eigleft2, eqright - eps * eigright1, eqright + eps * eigright2};
```

We want to move away from each saddle point, so we choose time steps of the appropriate algebraic sign -- in some case we want to integrate forward in time, in other cases backward. Again this comes from hindsight.

In[479]:=
```
stepset = {-0.02, -0.02, 0.02, 0.02, -0.02, 0.02};
```

In[480]:=
```
arrowvec = {1 / 2};
```

In[481]:=
```
setcolor[{Red}];
```

In[482]:=
```
bothdirflag = False;
```

In[483]:=

```
saddleport = portrait[initset, 0.0, stepset, 400, 1, 2]
```

Out[483]=



We appear to have a new periodic solution -- the oval-like closed curve with the slope breaks at $x = \pm 1$. Actually it is not a periodic solution. It takes infinitely long for the system point to arrive at either of those points. This curve separates the periodic and non-periodic solutions. Solutions starting inside the oval are periodic, and those starting outside go off to infinity. Let's combine all of our graphs for this example.

In[484]:=
```
show[graph1, graph2, graph3, graph4, graph5, graph6, saddleport]
```

Out[484]=



At this point, our examples represent the following range of possibilities: (1) a system may have no periodic solutions; (2) all solutions of a system may be periodic; (3) the phase plane may be divided into two regions, in one of which all solutions are periodic and in the other no solutions are periodic. Although this may seem like a complete catalogue of possibilites, there is one more unusual and very important possibility, which is illustrated by our last example.

■ Example 4: Nonlinear System with an Isolated Periodic Solution

This example is taken from **Nonlinear Ordinary Differential Equations**, D.W. Jordan and P. Smith, second edition, Oxford Press 1987, p. 18. The equations are

$$\dot{x} = y, \ \dot{y} = -x - (x^2+y^2-1)\, y \quad .$$

We define this for DynPac.

In[485]:=
```
setstate[{x, y}]; setparm[{}];
```

In[486]:=
```
slopevec = {y, -x - y (x^2 + y^2 - 1)};
```

In[487]:=
```
sysname = "Limit Cycle";
```

We start by looking for equilibrium states.

In[488]:=
```
findpolyeq
```
Out[488]=
```
{{0, 0}}
```

There is an equilibrium at the origin. Although findpolyeq didn't find any more, that doesn't prove that there are no more. Fortunately in this case it is easy to show that the origin is the only equilibrium: from the first equation, we find that $y = 0$ for equilibrium, and then the second equation shows that x must be zero for equilibrium. We classify the equilibrium:

In[489]:=
```
classify2D[{0, 0}]
```
```
unstable - spiral
```

Now we get the eigenvalues.

In[490]:=
```
eigval[{0, 0}]
```
Out[490]=
$$\left\{(-1)^{1/3}, -(-1)^{2/3}\right\}$$

In[491]:=
```
ComplexExpand[%]
```
Out[491]=
$$\left\{\frac{1}{2} + \frac{i\sqrt{3}}{2}, \frac{1}{2} - \frac{i\sqrt{3}}{2}\right\}$$

As we expect from an unstable spiral, we get complex conjugate eigenvalues with a common positive real part.

We construct a few orbits, starting near the origin, integrating both directions in time.

In[492]:=
```
initset = {{0.01, 0}, {-0.01, 0}, {0, 0.01}, {0, -0.01}};
```
In[493]:=
```
t0 = 0.0; h = 0.04; nsteps = 400;
```
In[494]:=
```
setcolor[{Black}];
```
In[495]:=
```
arrowvec = {1 / 4};
```
In[496]:=
```
plrange = {{-2, 2}, {-2, 2}};
```

In[497]:=

```
graph1 = portrait[initset, t0, h, nsteps, 1, 2]
```

Out[497]=



The four curves spiral out from the origin as we expected. Something very unexpected though is the accumulation of the orbits on what appears to be the unit circle. Let's look at a few more orbits, this time starting outside of the circle.

In[498]:=

```
initset = {{2, 0}, {-2, 0}, {0, 2}, {0, -2}};
```

In[499]:=

```
t0 = 0.0; h = 0.04; nsteps = 200;
```

In[500]:=
```
graph2 = portrait[initset, t0, h, nsteps, 1, 2]
```
Out[500]=



More surprises. These orbits spiral inward, and also appear to accumulate on the unit circle. It is time that we try to find out exactly what is going on. To do this, we use a strategy that often helps in two-dimensional problems. We introduce the polar radial coordinate r by

$$r^2 = x^2 + y^2 \text{ , so that}$$

$$r\dot{r} \;=\; x\dot{x} \;+\; y\dot{y} \;= xy + y(-x - y[r^2 - 1]) = y^2(1 - r^2) \;\; .$$

This equation is the key to the pictures above. We see that for $r > 1$, $\dot{r} < 0$, for $r < 1$, $\dot{r} > 0$, and for $r = 1$, $\dot{r} = 0$. Thus orbits starting inside the unit circle will move outward toward it, and orbits starting outside the unit circle will move inward toward it. What about the circle itself? When $r = 1$, $\dot{r} = 0$, so a point on the circle stays on the circle. We now make a complete switch to polar coordinates:

$$x = r\cos(\theta), \;\; y = r\sin(\theta).$$

Then the original differential equations become

$$\dot{r} \;=\; r\sin^2(\theta)\,(1 - r^2) \;,\;\; \dot{\theta} = -1 + (1 - r^2)\sin(\theta)\cos(\theta) \;\; .$$

Now the significance of the unit circle becomes clear. A solution of the above equations is

$$r = 1, \theta = -t \;,\;\; \text{or} \;\; x = \cos(t) \text{ and } y = -\sin(t) \;\; .$$

Thus the unit circle is the orbit of a periodic solution. The orbit is traversed clockwise, and the period is 2π. We may summarize the situation as follows: (1) this system has a single isolated periodic solution; (2) the periodic solution is an attractor, and every initial condition leads to the periodic solution. Such an isolated periodic solution is called a **limit cycle**. We will meet many more limit cycles in our furthur work. We combine our two graphs to show better the attractive nature of the limit cycle.

```
In[501]:=
      show[graph1, graph2]
```

```
Out[501]=
```



Can we construct the pure limit cycle without the approach curves? In this particular case it is easy because we happen to have a formula for the solution. A general approach, which works when we don't have such a formula, is to start more or less anywhere in the basin of attraction of the limit cycle and do a long integration. Then throw away the all of the results of that integration except the very last point, which is presumably on the limit cycle if we have integrated over long enough time. Use that last point as an intial condition and integrate again. Let's try it.

```
In[502]:=
      sol1 = integrate[{0.5, 0}, 0.0, 0.02, 500];
```

The last point of any integration is stored in the variable lastx:

```
In[503]:=
      lastx
```

```
Out[503]=
      {-0.693492, 0.720318}
```

We use this as an initial condition and integrate again.

```
In[504]:=
      sol2 = integrate[lastx, 0.0, 0.02, 500];
```

In[505]:=

```
phaser[sol2]
```

Out[505]=



We see this has worked quite well and we have the pure limit cycle. We can automate this process by using the DynPac function limcyc, which essentially does the same thing we have just done.

In[506]:=
```
phaser[limcyc[{0.5, 0}, 0.0, 0.02, 500]]
```
Out[506]=



For interpretive purposes, it is useful to put the original *x*-*y* system in the form of a single second order equation for *x*. The result is

$$\ddot{x} \;+\; (x^2+\dot{x}^2 - 1)\dot{x} + x = 0\ .$$

Now we see that this is an oscillator with a linear spring force and nonlinear damping. We also see that for $x^2+\dot{x}^2 < 1$, the damping is negative and the motion is amplified. For $x^2+\dot{x}^2 > 1$, the damping is positive, and the motion will decay. Thus in both cases the damping term acts to drive the system to the limit cycle. We shall see later that this limit cycle is atypically simple, but in general the concepts developed here apply to other limit cycles.

▪ Summary

Based on our four examples, we may summarize the possibilities for periodic solutions as follows: (1) there may be no periodic solutions; (2) all solutions may be periodic; (3) there may be a subset of phase space in which all solutions are periodic, with no periodic solutions outside this subset; (4) we may have one (or more) isolated periodic solutions. This is not exhaustive in the sense that combinations can occur. For example, one can have a system for which all solutions in some region are periodic, and for which there is an isolated periodic solution outside that region. It is also possible for a system to have infinitely many isolated periodic solutions. We shall see examples of all of these later.